
3. Correction to eMBMS-SC test cases 17.4.x
Change 1 – Correction to function ‘f_TC_17_4_1_Common’
	Function name
	f_TC_17_4_1_Common ()

(Note: TC 17.4.1. used as example for all eMBMS-SC test cases.)

	Reason for change
	See CR cover page.

	Summary of change
	Defined an “altstep” which handles the optional MBMS Interest Indication RRC message omitting the mbms-FreqList and activated this “altstep” in the postamble of affected eMBMS-SC test cases.

	TTCN module
	LTE-A\17\MBMS_SC_TestCases.ttcn

	MCC160 Comment
	


Before change
original TTCN source code of the “MBMS_SC_TestCases.ttcn” module in ATS_15wk38
After change

...
/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2015, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
//   All rights reserved.
// @version: IWD_15wk38
// $Date: 2015-09-15 19:05:27 +0200 (Tue, 15 Sep 2015) $
// $Rev: 14431 $
/******************************************************************************/
module MBMS_SC_TestCases {
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from CommonDefs all;
  import from EUTRA_CellInfo all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_Component all;
  import from EUTRA_BandDependentParam all;
  import from EPS_NAS_TypeDefs all;
  import from EUTRA_ASP_SrbDefs all;
  import from EPS_NAS_MsgContainers all;
  import from EUTRA_Measurement_Templates all;
  import from EUTRA_SRB_Templates all;
  import from EUTRA_RRC_Templates all;
  import from EUTRA_RRC_Templates all;
  import from EPS_NAS_Constants all;
  import from EPS_NAS_Templates all;
  import from EUTRA_AspCommon_Templates all;
  import from EUTRA_ConfigurationSteps all;
  import from EUTRA_CommonProcedures all;
  import from EUTRA_RRCSteps all;
  import from UpperTesterFunctions all;
  import from EUTRA_SecurityFunctions all;
  import from EUTRA_SecuritySteps all;
  import from EUTRA_LoopBack all;
  import from EUTRA_NASSteps all;
  import from EUTRA_Paging all;
  import from EUTRA_CommonTemplates_MBMS all;
  import from EUTRA_CommonFunctions_MBMS all;
  import from EUTRA_CellInfo_MBMS all;
  //=================================================================================
  // Templates
  //=================================================================================
  template (value) MeasConfig cs_Measurement_Config_A3_InterFreq_17_4_x(ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq1,
                                                                        AllowedMeasBandwidth p_MeasurementBandwidth1,
                                                                        ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq2,
                                                                        AllowedMeasBandwidth p_MeasurementBandwidth2,
                                                                        MeasObjectId p_MeasObjectId_2,
                                                                        template (value) MeasGapConfig p_MeasGapConfig := cs_508_MeasGapConfig_GP1, // @sic R5-153741 sic@
                                                                        template (omit) NeighCellConfig p_NeighCellConfig := '01'B) :=
    /* @status    APPROVED (LTE_A) */
    cs_MeasConfig(omit,
                  cs_MeasObjectToAddModList_2Entries(tsc_IdMeasObject_f1,
                                                     cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq1, p_MeasurementBandwidth1, -, -, -, -, -, p_NeighCellConfig),  // @sic R5-153741 sic@
                                                     p_MeasObjectId_2,
                                                     cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq2, p_MeasurementBandwidth2, -, -, -, -, -, p_NeighCellConfig)),  // @sic R5-153741 sic@
                  omit,
                  cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A3, cs_508_ReportConfigEUTRA_A3),
                  omit,
                  cs_MeasIdToAddModList_1Entry( 1, p_MeasObjectId_2, tsc_IdReportConfig_A3 ),
                  -,
                  p_MeasGapConfig);
  //=================================================================================
  //  TEST CASES
  //=================================================================================
  /*
   * @desc      To receive the MBMS interest indication in a parallel behaviour
   * @param     p_CellId
   * @status    APPROVED (LTE_A)
   */
  altstep a_MBMS_ReceiveMBMSInterestInd(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    //@sic R5s140757 R5s1401018 sic@
    var Frequency_IE_Type v_Frequency_CellId  := f_EUTRA_CellInfo_GetFrequencyIEs(p_CellId);
    var SRB_COMMON_IND v_ReceivedASP;
    [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({v_Frequency_CellId .UL_DL_Earfcn.dl_CarrierFreq})))) -> value v_ReceivedASP
      {
        f_EUTRA_MsgReceivedInDefault_SetSRB(v_ReceivedASP); //@sic R5s141018 sic@
        repeat;
      }
  }
  altstep a_MBMS_ReceiveMBMSInterestIndMbmsEnd(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({}))))
      {
        repeat;
      }
  }
  //=================================================================================
  //  TEST CASES
  //=================================================================================
  /*
   * @desc      Cell reselection to intra-frequency cell to continue MBMS service reception
   * @param     p_CellId
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_1_Common(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    var integer v_MBMS_CounterValue, v_MBMS_CounterValue2;
    var Frequency_IE_Type v_Frequency_Cell1;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(eutra_Cell11, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(p_CellId, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell11, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(p_CellId, -79)
    };
    var default v_DefaultRef;
    f_EUTRA_Init(c3);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c16);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell11,c16);
    // Cell 1 and  Cell 11 are part of the same MBSFN area @sic R5s140757 sic@
    f_EUTRA_InitialiseWithSameCellTiming ( {eutra_Cell1, eutra_Cell11} );
    if (p_CellId == eutra_Cell10) { //Test case 17.4.1a
      f_EUTRA_InitSIB5_Multiband (eutra_Cell1);
      f_EUTRA_InitSIB5_Multiband (p_CellId);
    }
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(eutra_Cell11);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell11); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellInfo_InitMaxReferencePower(p_CellId, -79);  //@sic R5s140757 sic@
    f_EUTRA_CellConfig_Def(p_CellId);
    v_Frequency_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeC);
    f_EUTRA_RbEst_Def(eutra_Cell1); // Default RB establishment
    f_EUTRA_CloseUE_TestLoopModeC(eutra_Cell1, cs_UE_TestLoopModeC_LB_Setup); // Bring UE in test loop mode C
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true,tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 0A-OC" siclog@
    f_ModifySysinfoCombinationC16_ToC20(eutra_Cell1);
    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);
    f_ModifySysinfoCombinationC16_ToC20(eutra_Cell11);
    f_EUTRA_ModifySysinfo(eutra_Cell11, false, RRC_CONNECTED);
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication( {v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq} ))));
    //@siclog "Step 1" siclog@
    //Release RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
    //@siclog "Steps 2-3" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data(eutra_Cell1, cs_MRB_Identity_Def,10, -,
                           rf8); //@sic R5s140026 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Step 4" siclog@
    //@sic R5s141160 sic@
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(eutra_Cell1));
    f_EUTRA_RbEst_Def(eutra_Cell1); // Default RB establishment
    deactivate(v_DefaultRef);
    //@siclog "Steps 5 - 7" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell1);
    if (v_MBMS_CounterValue > 0) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6: MBMS Counter value greater than 0");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6: MBMS Counter value not greater than 0");
    }
    //@siclog "Step 6a" siclog@
    //Release RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);
    //@siclog "Step 7" siclog@
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 8 Void" siclog@
    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.7 indicate that the UE is camped on E-UTRAN Cell 10?
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(eutra_Cell11));
    f_EUTRA_508CheckCampOnNewEutraCell( eutra_Cell11 );
    deactivate(v_DefaultRef);
    //@siclog "Step 9 Void" siclog@
    //@siclog "Step 10" siclog@
    f_Delay (12.0);
    //@siclog "Steps 11-12" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data (eutra_Cell11, cs_MRB_Identity_Def,10, -,
                            rf8); //@sic R5s140026 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Step 12A" siclog@
    //@sic R5s141160 sic@
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(eutra_Cell11));
    f_EUTRA_RbEst_Def(eutra_Cell11); // Default RB establishment
    deactivate(v_DefaultRef);
    //@siclog "Steps 13 - 15" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue2:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell11); //@sic R5s141358 sic@
    if (v_MBMS_CounterValue2 > v_MBMS_CounterValue) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15: MBMS Counter value greater than in step 5");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15: MBMS Counter value not greater than in step 5");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell11));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell11, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.1
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_1_EUTRA () runs on EUTRA_PTC
  { /* Cell reselection to intra-frequency cell to continue MBMS service reception */
    f_TC_17_4_1_Common( eutra_Cell3);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.1a
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_1a_EUTRA () runs on EUTRA_PTC
  { /* Cell reselection to intra-frequency cell to continue MBMS service reception inter- band */
    f_TC_17_4_1_Common( eutra_Cell10);
  }
  /*
   * @desc      Cell reselection to  inter-frequency/Inter Band cell to start MBMS service reception
   * @param     p_CellId
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_2_Common(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    var integer v_MBMS_CounterValue;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(p_CellId, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var Frequency_IE_Type v_Frequency_IE_Cell;
    var default v_DefaultRef := null;
    f_EUTRA_Init(c3);
    f_EUTRA_SetSysinfoCombination_MBMS(p_CellId,c16);
    //TAC for Cell 1 is set to 1 in f_EUTRA_Init
    if (p_CellId == eutra_Cell10) { //Test case 17.4.2a
      f_EUTRA_CellInfo_SetTAC(p_CellId, int2bit(2, 16));
      f_EUTRA_InitSIB5_Multiband (eutra_Cell1); //@sic R5s140840 sic@
      f_EUTRA_InitSIB5_Multiband (p_CellId);
    }
    f_EUTRA_InitialiseWithSameCellTiming ( {eutra_Cell1, p_CellId} ); // @sic R5s140840 sic@
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(p_CellId);
    f_EUTRA_CellConfig_MBMS_Def(p_CellId); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeC);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true,tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def);
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    v_Frequency_IE_Cell := f_EUTRA_CellInfo_GetFrequencyIEs(p_CellId);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 0A-OC" siclog@
    f_ModifySysinfoCombinationC3_ToC18(eutra_Cell1, {cs_508_MBMS_SAI_InterFreq_r11_Def(v_Frequency_IE_Cell.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                       tsc_MBMS_SAI_Def)},-,-,true); //@sic R5s140840 R5s141364 sic@
    f_EUTRA_ModifySysinfo(eutra_Cell1, true);
    f_ModifySysinfoCombinationC16_ToC20(p_CellId);
    f_EUTRA_ModifySysinfo(p_CellId, true);
    //@siclog "Step 1" siclog@
    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.7 indicate that the UE is camped on E-UTRAN Cell 10?
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(p_CellId));
    f_EUTRA_508CheckCampOnNewEutraCell( p_CellId );
    deactivate(v_DefaultRef); //@sic R5s140840 sic@
    //@siclog "Step 2 Void" siclog@
    //@siclog "Step 3" siclog@
    f_Delay (12.0);
     //@siclog "Step 4" siclog@
     v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(p_CellId));//@sic R5s150089 R5-150673 sic@
    f_EUTRA_RbEst_Def(p_CellId); // Default RB establishment
    deactivate(v_DefaultRef); //@sic R5s150089 R5-150673 sic@
    f_EUTRA_CloseUE_TestLoopModeC(p_CellId, cs_UE_TestLoopModeC_LB_Setup); // Bring UE in test loop mode C
    //@siclog "Step 4A" siclog@
    f_EUTRA_RRC_ConnectionRelease(p_CellId);
    //@siclog "Steps 5-6" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data(p_CellId, cs_MRB_Identity_Def, 10, -, rf8); //@sic R5s140840 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Step 6A" siclog@
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(p_CellId));//@sic R5s150089 R5-150673 sic@
    f_EUTRA_RbEst_Def(p_CellId); // Default RB establishment
    deactivate(v_DefaultRef); //@sic R5s150089 R5-150673 sic@
    //@siclog "Steps 7 - 9" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue:=f_EUTRA_MBMS_CounterProcedure(p_CellId);
    if (v_MBMS_CounterValue > 0) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9: MBMS Counter value greater than 0");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9: MBMS Counter value not greater than 0");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(p_CellId));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def );
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(p_CellId, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.2
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_2_EUTRA () runs on EUTRA_PTC
  { /* Cell reselection to inter-frequency cell to start MBMS service reception */
    f_TC_17_4_2_Common(eutra_Cell23); //@sic R5s140840 sic@
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.2a
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_2a_EUTRA() runs on EUTRA_PTC
  { /* Cell reselection to inter-band cell to start MBMS service reception */
    f_TC_17_4_2_Common(eutra_Cell10);
  }
  /*
   * @desc      Handover to inter-frequency/Inter Band cell to start MBMS service reception
   * @param     p_CellId
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_3_Common(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var Frequency_IE_Type v_Frequency_Cell1;
    var Frequency_IE_Type v_Frequency_CellX;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_CellX;
    var PhysCellId v_PhysCellId_CellX;
    var integer v_MBMS_CounterValue;
    var MeasObjectId v_MeasObjectId;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(p_CellId, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(p_CellId, -79)
    };
    var default v_DefaultRef;
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c16);
    f_EUTRA_SetSysinfoCombination_MBMS(p_CellId,c16);
    //@sic R5s150230 sic@
    if (p_CellId == eutra_Cell10) { //Test case 17.4.3a
      f_EUTRA_InitSIB5_Multiband (eutra_Cell1);
      f_EUTRA_InitSIB5_Multiband (p_CellId);
    }
     //@sic R5-153741 sic@
     f_EUTRA_CellInfo_SetSib3_NeighCellConfig ( eutra_Cell1,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
     f_EUTRA_CellInfo_SetSib5_NeighCellConfig ( eutra_Cell1,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
     f_EUTRA_CellInfo_SetSib3_NeighCellConfig ( p_CellId,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
     f_EUTRA_CellInfo_SetSib5_NeighCellConfig ( p_CellId,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
    // Cell 1 and  p_CellId are part of the same MBSFN area @sic R5s140757 sic@
    f_EUTRA_InitialiseWithSameCellTiming ( {eutra_Cell1, p_CellId} );
    f_EUTRA_CellInfo_InitMaxReferencePower(p_CellId, -79);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(p_CellId);
    f_EUTRA_CellConfig_MBMS_Def(p_CellId); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeC);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true,tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_RbEst_Def(eutra_Cell1); // Default RB establishment
    f_EUTRA_CloseUE_TestLoopModeC(eutra_Cell1, cs_UE_TestLoopModeC_LB_Setup); // Bring UE in test loop mode C
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    v_PhysCellId_CellX := f_EUTRA_CellInfo_GetPhyCellId(p_CellId);
    v_Frequency_CellX := f_EUTRA_CellInfo_GetFrequencyIEs(p_CellId);
    v_ChBandDependency_CellX := f_EUTRA_BandDependentParam(v_Frequency_CellX.DL_ChBandwidth, v_Frequency_CellX.UL_ChBandwidth);
    v_Frequency_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_Cell1.DL_ChBandwidth, v_Frequency_Cell1.UL_ChBandwidth);
    if (p_CellId == eutra_Cell3) { ///17.4.3
      v_MeasObjectId := tsc_IdMeasObject_f2;
    } else { //17.4.3a
      v_MeasObjectId := tsc_IdMeasObject_f5;
    }
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 0-1" siclog@
    f_ModifySysinfoCombinationC16_ToC20(eutra_Cell1, {cs_508_MBMS_SAI_InterFreq_r11_Def(v_Frequency_CellX.UL_DL_Earfcn.dl_CarrierFreq, 1)}, -, 2);//@sic R5s140821 sic@
    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);
    f_ModifySysinfoCombinationC16_ToC20(p_CellId, {cs_508_MBMS_SAI_InterFreq_r11_Def(v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq, 2)}, -, 1); //@sic R5s140821 sic@
    //@siclog "Step 1" siclog@
    f_EUTRA_ModifySysinfo(p_CellId, true, RRC_CONNECTED);
    //@siclog "Step 2" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({v_Frequency_CellX.UL_DL_Earfcn.dl_CarrierFreq}))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2: Does the UE transmit MBMSInterestIndication message");
    //@siclog "Steps 3-4" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup intra frequency measurement and
    //receives an RCConnectionReconfigurationComplete message.
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,
                                                                                                      cs_Measurement_Config_A3_InterFreq_17_4_x(v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                                                                                v_ChBandDependency_Cell1.AllowedMeasBandwidth,
                                                                                                                                                v_Frequency_CellX.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                                                                                v_ChBandDependency_CellX.AllowedMeasBandwidth,
                                                                                                                                                v_MeasObjectId,
                                                                                                                                                -,
                                                                                                                                                tsc_NeighCellConfig_NeighbourMBSFN))); //@sic R5-153741 sic@
    //@siclog "Step 5" siclog@
    //The SS changes Cell 1 and Cell 2 level according to the row "T1" in table 17.4.5.3.2-1
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 6" siclog@
    // The UE transmits a MeasurementReport message to report event A3 on Cell 1 with the measured RSRP, RSRQ value for Cell 2.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultListEUTRA_1Entry(v_PhysCellId_CellX, omit, ?, ?))));
    //@siclog "Steps 7-8" siclog@
    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra-frequency handover to Cell 2 less than [600ms] after reception of the MBMSInterestIndication message in step 8 (Note1).
    // The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2?
    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell1, p_CellId);
    //@siclog "Steps 9-10" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data(p_CellId, cs_MRB_Identity_Def,
                           10,
                           -,
                           rf8); //@sic R5s140026 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Steps 11 - 12" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue:=f_EUTRA_MBMS_CounterProcedure(p_CellId);
    if (v_MBMS_CounterValue > 0) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12: MBMS Counter value greater than 0");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 12: MBMS Counter value not greater than 0");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(p_CellId));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def );
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(p_CellId, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.3
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_3_EUTRA () runs on EUTRA_PTC
  { /* Handover to inter-frequency cell to start MBMS service reception */
    f_TC_17_4_3_Common( eutra_Cell3); //@sic R5s141352 sic@
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.3a
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_3a_EUTRA () runs on EUTRA_PTC
  { /* Handover to inter-band cell to start MBMS service reception */
    f_TC_17_4_3_Common( eutra_Cell10);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.4
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_4_EUTRA() runs on EUTRA_PTC
  { /* Handover to intra-frequency cell to continue MBMS service reception */
    var Frequency_IE_Type v_Frequency_Cell1, v_Frequency_Cell2;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;
    var PhysCellId v_PhysCellId_Cell2;
    var template (value) MeasConfig v_MeasConfig;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    var integer v_MBMS_CounterValue, v_MBMS_CounterValue2;
    var default v_DefaultRef;
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1, c15);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell2, c15);
    // Cell 1 and  Cell 2 are part of the same MBSFN area @sic R5s140757 sic@
    f_EUTRA_InitialiseWithSameCellTiming({eutra_Cell1, eutra_Cell2});
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell2); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeC);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_RbEst_Def(eutra_Cell1); // Default RB establishment
    f_EUTRA_CloseUE_TestLoopModeC(eutra_Cell1, cs_UE_TestLoopModeC_LB_Setup); // Bring UE in test loop mode C
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Steps 1" siclog@
    // The SS transmits a Paging message including a systemInfoModification for cell1
    // From the beginning of the next modification period the SS starts broadcast of SystemInformationBlockType15 (according to System information combination 19) on Cell 1 including mbms-SAI-IntraFreq-r11 list indicating MBMS SAI=1.
    //@siclog "Step 1-1A" siclog@
    f_ModifySysinfoCombinationC15_ToC19 (eutra_Cell1);
    f_ModifySysinfoCombinationC15_ToC19(eutra_Cell2);
    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);
    f_EUTRA_ModifySysinfo(eutra_Cell2, true, RRC_CONNECTED);
    //@siclog "Step 1B" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message.
    v_Frequency_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication( { v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq} ))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 1B: Does the UE transmit MBMSInterestIndication message?");
    //@siclog "Steps 2-3" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data (eutra_Cell1, cs_MRB_Identity_Def,10, -,
                            rf8); //@sic R5s140026 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Steps 4 - 6" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell1);
    if (v_MBMS_CounterValue > 0) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6: MBMS Counter value greater than 0");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6: MBMS Counter value not greater than 0");
    }
    //@siclog "Steps 7-8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to setup intra frequency measurement on Cell 2.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 2 to confirm the setup of intra frequency measurement.
    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    v_Frequency_Cell2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_Cell2.DL_ChBandwidth, v_Frequency_Cell2.UL_ChBandwidth);
    v_MeasConfig := cs_MeasConfig(omit,
                                  cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                   cs_MeasObject_measObjectEUTRA_Common(v_Frequency_Cell2.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency_Cell2.AllowedMeasBandwidth)),
                                  omit,
                                  cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A3,
                                                                     cs_508_ReportConfigEUTRA_A3),
                                  omit,
                                  cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A3));
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(tsc_RRC_TI_Def,v_MeasConfig));
    //@siclog "Step 9" siclog@
    //The SS changes Cell 1 and Cell 2 level according to the row "T1" in table 17.4.6.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 10" siclog@
    //The UE transmits a MeasurementReport message to report event A3 on Cell 1 with the measured RSRP, RSRQ value for Cell 2.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultListEUTRA_1Entry(v_PhysCellId_Cell2, omit, ?, ?))));
    //@siclog "Steps 11-12" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-frequency handover to Cell 2.
    //The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2
    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell1, eutra_Cell2);
    ///@siclog "Steps 13-14" siclog@
    // Send 10 MTCH packets
    f_EUTRA_Send_MTCH_Data (eutra_Cell2, cs_MRB_Identity_Def,10, -,
                            rf8); //@sic R5s140026 sic@
    f_Delay (6.0); //> 300ms + 15 * 320ms
    //@siclog "Steps 15 - 17" siclog@
    // Execute MBMS counter check procedure
    v_MBMS_CounterValue2:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell2);
    if (v_MBMS_CounterValue2 > v_MBMS_CounterValue) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17: MBMS Counter value greater than 0");
    } else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 17: MBMS Counter value not greater than 0");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell2));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def );
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.5
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_5_EUTRA() runs on EUTRA_PTC
  { /* Conditional retransmission of MBMS Interest Indication after handover */
    var integer v_MCCH_RepPer;
    var float v_Wait;
    var GutiParameters_Type v_Guti_Params;
    var Frequency_IE_Type v_Frequency_Cell2;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;
    var PhysCellId v_PhysCellId_Cell2;
    var template (value) MeasConfig v_MeasConfig;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell2, -79)
    };
    var default v_DefaultRef;
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c19);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell2,c19);
    //Cell 1 and Cell 2 belong to same MBSFN area. @sic R5s140757 sic@
    f_EUTRA_InitialiseWithSameCellTiming ( {eutra_Cell1, eutra_Cell2} );
    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);
    //Initialise SIB3 for cell1 and cell2 acc to Table 17.4.5.3.3-1A: SystemInformationBlockType3 for Cells 1 and 2
    //@sic R5-153740 sic@
    f_EUTRA_CellInfo_SetSib3_NeighCellConfig ( eutra_Cell1,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
    f_EUTRA_CellInfo_SetSib3_NeighCellConfig ( eutra_Cell2,
                                               tsc_NeighCellConfig_NeighbourMBSFN);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell2); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    //The UE is made aware that the MBMS service is active
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1" siclog@
    //Wait for a period equal to the MCCH repetition period for the UE to receive MBSFNAreaConfiguration message
    v_MCCH_RepPer := f_EUTRA_Convert_MCCH_RepPeriodToInt(f_EUTRA_CellInfo_GetMCCH_RepPeriod(eutra_Cell1));
    v_Wait := int2float(v_MCCH_RepPer)/100.0; // Rep Period is in frame = 10 ms
    f_Delay(v_Wait);
    //@siclog "Step 2" siclog@
    //The generic procedures described in TS 36.508 sub clause 4.5.3.3 are performed on Cell 1
    f_EUTRA_RbEst_Def(eutra_Cell1);
    //@siclog "Steps 3-4" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup intra frequency measurement and
    //receives an RCConnectionReconfigurationComplete message.
    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    v_Frequency_Cell2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_Cell2.DL_ChBandwidth, v_Frequency_Cell2.UL_ChBandwidth);
    //@sic R5-153740 R5s150764 sic@
     v_MeasConfig := cs_MeasConfig(omit,
                                  cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                   cs_MeasObject_measObjectEUTRA_Common(v_Frequency_Cell2.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency_Cell2.AllowedMeasBandwidth,omit,omit,omit,omit,omit,tsc_NeighCellConfig_NeighbourMBSFN)),
                                  omit,
                                  cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A3,
                                                                     cs_508_ReportConfigEUTRA_A3),
                                  omit,
                                  cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A3));
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(tsc_RRC_TI_Def,v_MeasConfig));
    //@siclog "Step 5" siclog@
    //The SS changes Cell 1 and Cell 2 level according to the row "T1" in table 17.4.5.3.2-1
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 6" siclog@
    // The UE transmits a MeasurementReport message to report event A3 on Cell 1 with the measured RSRP, RSRQ value for Cell 2.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultListEUTRA_1Entry(v_PhysCellId_Cell2, omit, ?, ?))));
    //@siclog "Step 7" siclog@
    // The UE is made interested in receiving MBMS service with MBMS Service ID=0 associated with the MBMS SAI (1) broadcasted in SIB15 mbms-SAI-IntraFreq list on Cell 1 and Cell 2
    v_Guti_Params := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def);
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({v_Frequency_Cell2.UL_DL_Earfcn.dl_CarrierFreq}))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8: Does the UE transmit MBMSInterestIndication message?");
    //@siclog "Steps 9-10" siclog@
    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra-frequency handover to Cell 2 less than [600ms] after reception of the MBMSInterestIndication message in step 8 (Note1).
    // The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2?
    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell1, eutra_Cell2);
    //@siclog "Step 11" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({ v_Frequency_Cell2.UL_DL_Earfcn.dl_CarrierFreq}))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11: Does the UE transmit MBMSInterestIndication message?");
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell2));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def);
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.6
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_6_EUTRA() runs on EUTRA_PTC
  { /* MBMS Interest Indication retransmission after returning from cell not broadcasting SIB15 */
    var Frequency_IE_Type v_Frequency_Cell1;
    var Frequency_IE_Type v_Frequency_Cell2;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;
    var PhysCellId v_PhysCellId_Cell1;
    var PhysCellId v_PhysCellId_Cell2;
    var template (value) MeasConfig v_MeasConfig;
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT2 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    var default v_DefaultRef;
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c15);
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    //Perform Idle updated procedure to make sure UE is in a known state before test case starts
    //Bring UE to initial state
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    //UE is in state Generic RB Established (state 3) on Cell 1 according to [18].
    f_EUTRA_RbEst_Def(eutra_Cell1);
    //The UE is made interested in receiving a MBMS service with MBMS Service ID=0 associated with the MBMS SAI (1) broadcasted in SIB15 mbms-SAI-IntraFreq list on Cell 1
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def);
    //The UE is made aware that the MBMS service is active
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Steps 0-1" siclog@
    // The SS transmits a Paging message including a systemInfoModification for cell1
    // From the beginning of the next modification period the SS starts broadcast of SystemInformationBlockType15 (according to System information combination 19) on Cell 1 including mbms-SAI-IntraFreq-r11 list indicating MBMS SAI=1.
    f_ModifySysinfoCombinationC15_ToC19(eutra_Cell1);
    f_EUTRA_ModifySysinfo( eutra_Cell1, true, RRC_CONNECTED );
    //@siclog "Step 2" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message.
    v_Frequency_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication( { v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq} ))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2: Does the UE transmit MBMSInterestIndication message?");
    //@siclog "Steps 3-4" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to setup intra-frequency measurement.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1 to confirm the setup of intra-frequency measurement.
    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    v_Frequency_Cell2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_Cell2.DL_ChBandwidth, v_Frequency_Cell2.UL_ChBandwidth);
    v_MeasConfig := cs_MeasConfig(omit,
                                  cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                   cs_MeasObject_measObjectEUTRA_Common(v_Frequency_Cell2.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency_Cell2.AllowedMeasBandwidth)),
                                  omit,
                                  cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A3,
                                                                     cs_508_ReportConfigEUTRA_A3),
                                  omit,
                                  cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A3));
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(tsc_RRC_TI_Def,v_MeasConfig));
    //@siclog "Step 5" siclog@
    //The SS changes Cell 1 and Cell 2 level according to the row "T1" in table 17.4.6.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 6" siclog@
    //The UE transmits a MeasurementReport message to report event A3 on Cell 1 with the measured RSRP, RSRQ value for Cell 2.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultListEUTRA_1Entry(v_PhysCellId_Cell2, omit, ?, ?))));
    //@siclog "Steps 7-8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-frequency handover to Cell 2.
    //The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2
    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell1, eutra_Cell2);
    //@siclog "Steps 9-10" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to setup intra frequency measurement on Cell 2.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 2 to confirm the setup of intra frequency measurement.
    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);
    v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_Cell1.DL_ChBandwidth, v_Frequency_Cell1.UL_ChBandwidth);
    v_MeasConfig := cs_MeasConfig(omit,
                                  cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,
                                                                   cs_MeasObject_measObjectEUTRA_Common(v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq, v_ChBandDependency_Cell1.AllowedMeasBandwidth)),
                                  omit,
                                  cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_A3,
                                                                     cs_508_ReportConfigEUTRA_A3),
                                  omit,
                                  cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_A3));  //@sic R5s140838 sic@
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell2,
                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(tsc_RRC_TI_Def,v_MeasConfig));
    //@siclog "Step 11" siclog@
    //The SS changes Cell 1 and Cell 2 levels according to the row "T2" in table 17.4.6.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    //@siclog "Step 12" siclog@
    //The UE transmits a MeasurementReport message to report event A3 on Cell 2 with the measured RSRP, RSRQ value for Cell 1.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2,
                                    cr_MeasurementReport(1, ?, ?, cr_MeasResultNeighCells_measResultListEUTRA_1Entry(v_PhysCellId_Cell1, omit, ?, ?)))); //@sic R5s140838 sic@
    //@siclog "Steps 13-14" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 2 to order the UE to perform intra-frequency handover to Cell 1.
    //The UE transmit an RRCConnectionReconfigurationComplete message on Cell 1?
    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell2, eutra_Cell1);
    //@siclog "Step 15" siclog@
    //Check: Does the UE transmit MBMSInterestIndication message.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication( { v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq} ))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15: Does the UE transmit MBMSInterestIndication message?");
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell1));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def );
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.7
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_7_EUTRA() runs on EUTRA_PTC
  { /* MBMS Interest Indication after Radio Link Failure */
    var EUTRA_SecurityParams_Type v_Auth_Params;
    var float v_ModificationPeriod_s; // Modification period in second
    var template (value) CellPowerList_Type v_CellPowerList_AtT0 := {
      cs_CellPower(eutra_Cell2, tsc_NonSuitableCellRS_EPRE)
    };
    var template (value) CellPowerList_Type v_CellPowerList_AtT1 := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    var default v_DefaultRef;
    //@sic R5s141018 sic@
    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;
    var template (omit) DRX_Config v_DRX_Config;
    var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_CellInfo_Cell2;
    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo_Cell2;
    var SR_PUCCH_ResourceIndex_Type v_SR_PUCCH_ResourceIndex_Cell2;
    var SR_ConfigIndex_Type v_SR_ConfigIndex_Cell2;
    var SRB_COMMON_IND v_ReceivedASP; //@sic R5s150090 sic@
    var Frequency_IE_Type v_Frequency_CellId;  //@sic R5s150090 sic@
    var GutiParameters_Type v_GutiParameters; //@sic R5s150075 sic@
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def; //@sic R5s150075 sic@
    var NAS_MSG_Indication_Type v_NasInd; //@sic R5s150075 sic@
    var NAS_KsiValue v_KsiValue; //@sic R5s150075 sic@
    timer t_Wait2S := 2.0;  //@sic R5s150075 R5-151779 sic@
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c15);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell2,c15);
    f_EUTRA_CellInfo_SetSysInfo_T310(eutra_Cell1, ms200);  //@sic R5s141018 sic@
    f_EUTRA_CellInfo_SetSysInfo_T310(eutra_Cell2, ms200);  //@sic R5s141018 sic@
    //Cell 1 and Cell 2 belong to same MBSFN area. @sic R5s140757 sic@
    f_EUTRA_InitialiseWithSameCellTiming ( {eutra_Cell1, eutra_Cell2} );
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell2); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //@sic R5s141018 sic@
    v_EUTRA_FDD_TDD_CellInfo_Cell2 := f_EUTRA_CellInfo_GetFDD_TDD_Info( eutra_Cell2);
    v_SR_PUCCH_ResourceIndex_Cell2 := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(eutra_Cell2);
    v_SR_ConfigIndex_Cell2 := f_EUTRA_CellInfo_GetSR_ConfigIndex(eutra_Cell2);
    v_AntennaInfo_Cell2 := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell2);
    //Bring UE to initial state
    //The UE is in state Registered, Idle Mode (state 2) on Cell 1 according to [18].
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    //TThe UE is made interested in receiving MBMS service with MBMS Service ID=0 with the MBMS SAI (1) broadcasted in SIB15 mbms-SAI-IntraFreq list on Cell 1 and Cell 2
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def);
    //The UE is made aware that the MBMS service is active.
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def), true, tsc_MBMS_SAI_Def );
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    f_EUTRA_TestBody_Set(true);
    //@siclog "Steps 0A-0B" siclog@
    // The SS transmits a Paging message including a systemInfoModification for Cell1 and Cell 2
    // From the beginning of the next modification period the SS starts broadcast of SystemInformationBlockType15 according to System information combination 19 as defined in TS 36.508[18] clause 4.4.3.1 on Cell 1 and Cell 2 including mbms-SAI-IntraFreq-r11 list indicating MBMS SAI=1.
    f_ModifySysinfoCombinationC15_ToC19(eutra_Cell1);
    f_EUTRA_ModifySysinfo( eutra_Cell1 ); //@sic R5s141018 sic@
    f_ModifySysinfoCombinationC15_ToC19(eutra_Cell2);
    f_EUTRA_ModifySysinfo( eutra_Cell2);  //@sic R5s150075 sic@
    //Wait 2.1*modification period to allow new system information to take effect
    v_ModificationPeriod_s := int2float(f_CalculateModificationPeriod (eutra_Cell1))/100.0;
    f_Delay (2.5*v_ModificationPeriod_s); //@sic R5s141018 R5s150075 sic@
    // receive any MBMSInterestIndi message
    // In parallel to the events described in step 1 the steps specified in Table 17.4.7.3.2-3 take place.
    //@sic R5s141018 sic@
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(eutra_Cell1));
    //@siclog "Steps 1-1D " siclog@
    //Generic test procedure Generic Radio Bearer Establishment as described in TS 36.508 subclause 4.5.3 is executed.
    //@sic R5s141018 sic@
    v_DRX_Config := f_GetUeDrxConfig(DRX_S);
    v_MAC_MainConfig := cs_MAC_MainConfig_Explicit(cs_MAC_MainConfig_Common(-, v_DRX_Config));
    //@sic R5s150075 sic@
    //Generic RB establishment procedure expended to be able to receive the MBMS interest indication before the security procedure
    v_GutiParameters := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    f_EUTRA_UE_Page(eutra_Cell1,
                    cs_NasPaging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_GutiParameters.MME_Code, v_GutiParameters.M_TMSI)));
    // Step 3 - 5 of Generic RB establishment procedure
    //Check the Service Request message
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_Cell1,
                                               v_RRC_TI,
                                               ?,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cr_508_SERVICE_REQUEST(v_KsiValue)));
    v_Frequency_CellId := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    //@siclog "Step 1E " siclog@
    t_Wait2S.start;   //@sic R5-151779 sic@
    alt {
        [] t_Wait2S.timeout {}
        []SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({v_Frequency_CellId.UL_DL_Earfcn.dl_CarrierFreq})))) -> value v_ReceivedASP
        {
            t_Wait2S.stop;
            f_EUTRA_MsgReceivedInDefault_SetSRB(v_ReceivedASP);
        }
    }
    //@siclog "Steps 1F-1J " siclog@
    /* Step 6 - 9 of Generic RB establishment procedure */
    f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                         tsc_RRC_TI_Def,
                                         v_NasInd.SecurityProtection.NasCount,
                                         0,
                                         0,
                                         -,
                                         -,    /* @sic R5s150338 sic@ */
                                         v_MAC_MainConfig);
    /* Step 9 - 10 of Generic RB establishment procedure */
    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def,
                                                  0,
                                                  0);
    //@siclog "Step 2" siclog@
    //UE transmits MBMSInterestIndication message.
    //@sic R5s141018 sic@
    deactivate(v_DefaultRef);
    if ( not(isvalue(f_EUTRA_MsgReceivedInDefault_Retreive()))) {
         f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2: UE did not transmit MBMSInterestIndication message") }
    else {
         f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2: UE transmitted MBMSInterestIndication message");
    }
    //@siclog "Step 3" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T1" in table 17.4.7.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    // Activate RRC security at SS target cell (Cell2) for SRB 1, 2 and 1 AM DRB
    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell2, f_EUTRA_Security_Get(), 0);
    f_EUTRA_Security_Set(v_Auth_Params);
    //@siclog "Step 4" siclog@
    //UE transmits RRCConnectionReestablishmentRequest message
    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell2,
                                    cr_508_RRCConnectionReestablishmentRequest(?,
                                                                               ?,
                                                                               ?,
                                                                               ?)));
    //@siclog "Step 5 " siclog@
    //SS transmits RRCConnectionReestablishment message.
    f_EUTRA_RRC_RRCConnectionReestablishment_Def(eutra_Cell2);
    //@siclog "Step 6" siclog@
    //UE transmits RRCConnectionReestablishmentComplete message.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReestablishmentComplete(tsc_RRC_TI_Def)));
    //@siclog "Steps 7-8" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1.
    //@sic R5s141018 sic@
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestInd(eutra_Cell2));
    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell2,
                                 cs_TimingInfo_Now,
                                 cs_RRCConnReconfig_Resume_SRB2_DRBs(tsc_RRC_TI_Def,
                                                                     v_EUTRA_FDD_TDD_CellInfo_Cell2.cqi_ReportConfig,
                                                                     v_EUTRA_FDD_TDD_CellInfo_Cell2.soundingRS_UL_Config,
                                                                     v_AntennaInfo_Cell2,
                                                                     v_SR_PUCCH_ResourceIndex_Cell2,
                                                                     v_SR_ConfigIndex_Cell2)));
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.
    //@sic R5s141354 sic@
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    deactivate(v_DefaultRef);
    //Check: Does the UE transmit MBMSInterestIndication message.
    // MBMSInterest Indication not yet received  @sic R5s150090 sic@
    if ( not(isvalue(f_EUTRA_MsgReceivedInDefault_Retreive()))) {
        v_Frequency_CellId := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
        SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication({v_Frequency_CellId.UL_DL_Earfcn.dl_CarrierFreq})))) -> value v_ReceivedASP;
        f_EUTRA_MsgReceivedInDefault_SetSRB(v_ReceivedASP);
        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7a: UE transmitted MBMSInterestIndication message");
    } else {
         f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7a: UE transmitted MBMSInterestIndication message");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell2));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),false, tsc_MBMS_SAI_Def );
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell2, E2_CONNECTED);
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 17.4.8
   * @status    APPROVED (LTE_A)
   */
  function f_TC_17_4_8_EUTRA() runs on EUTRA_PTC
  { /* Continued MBMS service reception after E-UTRAN release of unicast bearer */
    var Frequency_IE_Type v_Frequency_Cell1;
    var integer v_MBMS_CounterValue1 := 0;
    var integer v_MBMS_CounterValue2 := 0;
    var integer v_MCCH_RepPer;
    var float v_Wait;
    var ProcedureTransactionIdentifier v_EPS_TI := '00'O; //@sic R5s140836 sic@
    var default v_DefaultRef;
    f_EUTRA_Init(c1);
    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c15);
    f_EUTRA_CellConfig_Def(eutra_Cell1);        /* @sic R5s150338: f_EUTRA_CellConfig_DRB_Common replaced by f_EUTRA_CellConfig_Def sic@ */
    f_EUTRA_CellConfig_MBMS_Def(eutra_Cell1); // Configured MBSFN + MTCH + transmit MBSFNAreaConfiguration
    //Bring UE to initial state
    //UE is in state Loopback Activated (State 4) with UE TEST LOOP MODE C on Cell 1
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeC);
    f_EUTRA_RbEst_Def(eutra_Cell1); // Default RB establishment
    f_EUTRA_CloseUE_TestLoopModeC(eutra_Cell1, cs_UE_TestLoopModeC_LB_Setup); // Bring UE in test loop mode C
    //Activate the second DRB @sic R5s140836 sic@
    f_EUTRA_TxDedicatedEpsBearerCtxtReq(eutra_Cell1,
                                        tsc_DRB2,
                                        tsc_EpsDedicatedBearerId,
                                        v_EPS_TI,
                                        cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),
                                        cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508
                                        f_EUTRA_Get_508_TrafficFlowTemplate(1)); // According to Context#1 in TS 36.508
    //The UE transmits an ACTIVATE DEDICATE EPS BEARER CONTEXT ACCEPT message.
    //@sic R5s140836 sic@
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));
    //The UE is made interested in receiving MBMS service with MBMS Service ID=0 in the PLMN associated with the MBMS SAI (1) broadcasted in SIB15 on Cell 1.
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), true);
    //The UE is configured to prioritise reception of MBMS frequencies above reception of any unicast bearers
    f_UT_MBMS_Unicast( UT, true) ;
    //The UE is made aware that the MBMS service is active
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def),true, tsc_MBMS_SAI_Def );
    f_EUTRA_TestBody_Set(true);
    //@siclog "Steps 1-3" siclog@
    //The SS transmits a Paging message including a systemInfoModification for cell1.
    //From the beginning of the next modification period the SS transmits SystemInformationBlockType15 including mbms-SAI-IntraFreq-r11 list indicating MBMS SAI=1.
    f_ModifySysinfoCombinationC15_ToC19 (eutra_Cell1);
    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED );
    //Check: Does UE transmit a MBMSInterestIndication message including the mbms-Priority IE set to True?
    v_Frequency_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MbmsInterestIndication_Common(cr_MbmsInterestIndication( { v_Frequency_Cell1.UL_DL_Earfcn.dl_CarrierFreq}, true_ ))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2: Does the UE transmit MBMSInterestIndication message?");
    //@siclog "Step 4" siclog@
    //Wait for a period equal to the MCCH repetition period for the UE to receive MBSFNAreaConfiguration message
    v_MCCH_RepPer := f_EUTRA_Convert_MCCH_RepPeriodToInt(f_EUTRA_CellInfo_GetMCCH_RepPeriod(eutra_Cell1)); //@sic R5s140836 sic@
    v_Wait := int2float(v_MCCH_RepPer)/100.0; // Rep Period is in frame = 10 ms;
    f_Delay(v_Wait);
    //@siclog "Steps 6-7" siclog@
    //The SS transmits a valid MAC PDU including 'MCH Scheduling Information MAC Control Element with LCID='00001', Stop MTCH= '00000000001'
    //The SS transmits 10 MBMS Packets on the MTCH.
    f_EUTRA_Send_MTCH_Data(eutra_Cell1, cs_MRB_Identity_Def(tsc_Mbsfn_AreaId_Def), 10, -, rf8); // transmit 10 MTCH @sic R5s140192 R5s140836 sic@
    f_Delay(1.0); //> 300ms + 10 * 4  SFN
    //@siclog "Steps 8-9" siclog@
    //The SS transmits an UE TEST LOOP MODE C MBMS PACKET COUNTER REQUEST message.
    //UE responds with UE TEST LOOP MODE C MBMS PACKET COUNTER RESPONSE.
    v_MBMS_CounterValue1:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell1);
    //@siclog "Steps 10" siclog@
    //Check: Is the number of reported MBMS Packets received on the MTCH in step 9 greater than zero?
    if (v_MBMS_CounterValue1 > 0)   {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10: MBMS Counter value greater than 0");
    } else  {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10: MBMS Counter value not greater than 0");
    }
    //@siclog "Steps 11-12" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to release the unicast bearer due to congestion on the MBMS carrier(s)
    //The UE transmits an RRCConnectionReconfigurationComplete message to confirm the release.
    f_EUTRA_RRCConnectionReconfiguration_DRB_Release(eutra_Cell1, tsc_RRC_TI_Def, tsc_DRB2);
    //@siclog "Steps 13-14" siclog@
    //The SS transmits a valid MAC PDU including  'MCH Scheduling Information MAC Control Element with LCID='00001', Stop MTCH= '00000000001'
    //The SS transmits 10 MBMS Packets on the MTCH.
    f_EUTRA_Send_MTCH_Data(eutra_Cell1, cs_MRB_Identity_Def(tsc_Mbsfn_AreaId_Def), 10, -, rf8); // transmit 10 MTCH @sic R5s140192 R5s140836 sic@
    f_Delay(1.0); //> 300ms + 10 * 4  SFN
    //@siclog "Steps 15-16" siclog@
    //The SS transmits an UE TEST LOOP MODE C MBMS PACKET COUNTER REQUEST message.
    //UE responds with UE TEST LOOP MODE C MBMS PACKET COUNTER RESPONSE.
    v_MBMS_CounterValue2:=f_EUTRA_MBMS_CounterProcedure(eutra_Cell1);
     //@siclog "Steps 17" siclog@
    //Does the number of reported MBMS Packets received on the MTCH is greater than the number of MBMS Packets reported in step10?
    if (v_MBMS_CounterValue2>v_MBMS_CounterValue1)   {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17: ");
    } else  {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 17: ");
    }
    f_EUTRA_TestBody_Set(false);
    v_DefaultRef := activate (a_MBMS_ReceiveMBMSInterestIndMbmsEnd(eutra_Cell1));
    f_UT_MBMS_SERVICE_Active(UT, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    f_UT_MBMS_Unicast( UT, false) ;
    f_UT_MBMS_SERVICE_Interest(UT, omit, oct2str(tsc_ServiceId_Def), false, tsc_MBMS_SAI_Def);
    // Release RRC Connection and Switch off
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);  //@sic R5s140836 sic@
  }
}

